# Maintainer: Robin Candau <antiz@archlinux.org>
# Contributor: Mark Wagie <mark dot wagie at proton dot me>
# Contributor: librewish <librewish at gmail dot com>
# Contributor: Chris Severance aur.severach aATt spamgourmet dott com
# Contributor: Jay Garcia <morbidj at gmail dot com>
# Contributor: Doug Newgard <scimmia22 at outlook dot com>
# Contributor: Robert Orzanna <orschiro at gmail dot com>

pkgname=timeshift
pkgver=23.07.1
pkgrel=2
pkgdesc="A system restore utility for Linux"
arch=('loong64' 'x86_64')
url="https://github.com/linuxmint/timeshift"
license=('GPL2')
depends=('libgee' 'json-glib' 'gtk3' 'vte3' 'rsync' 'xapp' 'cronie' 'which')
makedepends=('help2man' 'meson' 'vala')
optdepends=('btrfs-progs: For BTRFS support'
            'grub-btrfs: For BTRFS snapshots in GRUB'
            'xorg-xhost: For authorization on Wayland')
install="${pkgname}.install"
source=("${pkgname}-${pkgver}.tar.gz::${url}/archive/refs/tags/${pkgver}.tar.gz")
sha256sums=('835277fa21182188278260d9690e067e9d3bacd6793ffdcbe89bb128ec672a01')

build() {
	arch-meson "${pkgname}-${pkgver}" build
	meson compile -C build
}

check() {
	meson test -C build
}

package() {
	meson install -C build --destdir "${pkgdir}"
}
